Problem 4.72 Using Excel as a calculator.

Description Propane is burned with air. Determine the molar composition of the product gas if no CO is formed, assuming:
(a) Theoretical air supplied, 100% conversion of fuel.
(b) 20% excess air, 100% conversion.
(c) 20% excess air, 90% conversion. Calculate the molar composition of the stack gas on a dry basis.

Solution:

For problem 4.72, parts (a) and (b) are essentially the same problem with a different value for excess air. Part (c) requires mole fraction calculation on a dry basis and with a different value for conversion. If we set up a spreadsheet with both wet and dry mole fractions and with chageable values for excess air and conversion, we can use one sheet to solve parts (a), (b), and (c).

First we draw the flowsheet and label the streams.

Flowsheet Drawing



Then we write down the balanced reaction equation: $C_3H_8 + 5O_2 \rightarrow 3CO_2 + 4H_2O$, or $P + 5O \rightarrow 3C + 4W$ with extent of reaction ξ.

We'll feed the air into Stream 1 and the propane into Stream 2.

Choose a basis, in this case we'll use 100 mol/s of P in the feed.

$$n_{2_P} = 100\frac{\text{mol}}{\text{s}}.$$

Then by the definition of excess air and the balanced reaction,

$$n_{1_O} = 5(1 + xs)n_{2_P}$$

where xs is the fraction of excess air.

For standard air,

$$n_{1_N} = \frac{79}{21} n_{1_O}.$$

Next, a mass balance for each species around the mixer yields,
$n_{3_i} = n_{1_i} + n_{2_i}$.

A mole balance on the reactor yields,
$n_{4_i} = n_{3_i} + \nu_i \xi$.

In terms of fractional conversion,

$$\xi = \frac{n_{1_P} f_P}{-(-1)}.$$

To calculate mole fractions in each stream we have

$$x_i = \frac{n_i}{\sum n_i}.$$

If the stream is on a dry basis we just leave water out of the calculation.
Nomenclature: In a spreadsheet, the cells containing entries that can be changed are in *italics*, the intermediate calculated quantities are in plain text, and the answers are in **bold text**.

To solve the problem, the following formulas are entered into a spreadsheet

Species	Stoic. Coeff	Stream 1	Stream 2	Stream 3	Stream 4		
C3H8	*-1*	*0*	*100*	=C2+D2	=E2+B2*B16		
O2	*-5*	=B3/B2*(1+B14)*D2	*0*	=C3+D3	=E3+B3*B16		
N2	*0*	=79/21*C3	*0*	=C4+D4	=E4+B4*B16		
CO2	*3*	*0*	*0*	=C5+D5	=E5+B5*B16		
H2O	*4*	*0*	*0*	=C6+D6	=E6+B6*B16		
	Total mole/s	=SUM(C2:C6)	=SUM(D2:D6)	=SUM(E2:E6)	=SUM(F2:F6)	**Dry Total**	=SUM(F2:F5)
	xC3H8	=C2/C$7	=D2/D$7	=E2/E$7	**=F2/F$7**	**xC3H8-dry**	**=F2/H$7**
	xO2	=C3/C$7	=D3/D$7	=E3/E$7	**=F3/F$7**	**xO2-dry**	**=F3/H$7**
	xN2	=C4/C$7	=D4/D$7	=E4/E$7	**=F4/F$7**	**xN2-dry**	**=F4/H$7**
	xCO2	=C5/C$7	=D5/D$7	=E5/E$7	**=F5/F$7**	**xCO2-dry**	**=F5/H$7**
	xH2O	=C6/C$7	=D6/D$7	=E6/E$7	**=F6/F$7**		

Frac. xs	*0*
f C3H8	*1*
extent	=D2*B15/(-B2)

If you are running this tutorial in *FrameMaker* and you have *Microsoft Excel* available, you can double-click the spreadsheets above and below and they will operate just like in *Excel*.

A major time saving trick is to enter a cell and then choose *Edit->Fill->Down* (Ctrl-d) or *Edit->Fill->Right* (Ctrl-r). For example in the above spreadsheet:



Species	Stoic. Coeff	Stream 1	Stream 2	Stream 3	Stream 4		
C3H8	*-1*	*0*	*100*	=C2+D2	=E2+B2*B16		
O2	*-5*	=B3/B2*(1+B14)*D2	*0*	=C3+D3	=E3+B3*B16		
N2	*0*	=79/21*C3	*0*	=C4+D4	=E4+B4*B16		
CO2	*3*	*0*	*0*	=C5+D5	=E5+B5*B16		
H2O	*4*	*0*	*0*	=C6+D6	=E6+B6*B16		
	Total mole/s	=SUM(C2:C6)	=SUM(D2:D6)	=SUM(E2:E6)	=SUM(F2:F6)	**Dry Total**	=SUM(F2:F5)
	xC3H8	=C2/C$7	=D2/D$7	=E2/E$7	**=F2/F$7**	**xC3H8-dry**	**=F2/H$7**
	xO2	=C3/C$7	=D3/D$7	=E3/E$7	**=F3/F$7**	**xO2-dry**	**=F3/H$7**
	xN2	=C4/C$7	=D4/D$7	=E4/E$7	**=F4/F$7**	**xN2-dry**	**=F4/H$7**
	xCO2	=C5/C$7	=D5/D$7	=E5/E$7	**=F5/F$7**	**xCO2-dry**	**=F5/H$7**
	xH2O	=C6/C$7	=D6/D$7	=E6/E$7	**=F6/F$7**		

Frac. xs	*0*
f C3H8	*1*
extent	=D2*B15/(-B2)

Fill right was used for **Total mole/s** (C7 to F7). Fill down was used for **Stream 3** (E2 to E6). The $'s and fill down were used for **Stream 4** (F2 to F6) and **x-dry** (H8 to H11). Using fill down, fill right, and $ can save you a lot of typing. For more information read "*About cell and range references*" and "*Cell references are wrong after I move or copy cells*" in *Excel online help*.

For part (a) the values are entered as follows:

Species	Stoic. Coeff	Stream 1	Stream 2	Stream 3	Stream 4		
C3H8	*-1*	*0*	*100*	100	0		
O2	*-5*	500	*0*	500	0		
N2	*0*	1880.952	*0*	1880.952	1880.952		
CO2	*3*	*0*	*0*	0	300		
H2O	*4*	*0*	*0*	0	400		
	Total mole/s	2380.952	100	2480.952	2580.952	**Dry Total**	2180.952
	xC3H8	0	1	0.040307	**0**	**xC3H8-dry**	**0**
	xO2	0.21	0	0.201536	**0**	**xO2-dry**	**0**
	xN2	0.79	0	0.758157	**0.728782**	**xN2-dry**	**0.862445**
	xCO2	0	0	0	**0.116236**	**xCO2-dry**	**0.137555**
	xH2O	0	0	0	**0.154982**		
Frac. xs	*0*						
f C3H8	*1*						
extent	100						

Again, the following key is used in the spreadsheet: Numbers that are entered or changeable by the user are in *italics*. Calculated intermediate values are in normal text and calculated values for the answer are in **bold**.

We can do part (b) by changing the fraction excess air from 0 to 0.2 (20% excess air)

Species	Stoic. Coeff	Stream 1	Stream 2	Stream 3	Stream 4		
C3H8	*-1*	*0*	*100*	100	0		
O2	*-5*	600	*0*	600	100		
N2	*0*	2257.143	*0*	2257.143	2257.143		
CO2	*3*	*0*	*0*	0	300		
H2O	*4*	*0*	*0*	0	400		
	Total mole/s	2857.143	100	2957.143	3057.143	**Dry Total**	2657.143
	xC3H8	0	1	0.033816	**0**	**xC3H8-dry**	**0**
	xO2	0.21	0	0.202899	**0.03271**	**xO2-dry**	**0.037634**
	xN2	0.79	0	0.763285	**0.738318**	**xN2-dry**	**0.849462**
	xCO2	0	0	0	**0.098131**	**xCO2-dry**	**0.112903**
	xH2O	0	0	0	**0.130841**		
Frac. xs	*0.2*						
f C3H8	*1*						
extent	100						

And we can do part c) by changing the conversion from 1 to 0.9 (100% to 90%)

Species	Stoic. Coeff	Stream 1	Stream 2	Stream 3	Stream 4		
C3H8	*-1*	*0*	*100*	100	10		
O2	*-5*	600	*0*	600	150		
N2	*0*	2257.143	*0*	2257.143	2257.143		
CO2	*3*	*0*	*0*	0	270		
H2O	*4*	*0*	*0*	0	360		
	Total mole/s	2857.143	100	2957.143	3047.143	**Dry Total**	2687.143
	xC3H8	0	1	0.033816	**0.003282**	**xC3H8-dry**	**0.003721**
	xO2	0.21	0	0.202899	**0.049226**	**xO2-dry**	**0.055821**
	xN2	0.79	0	0.763285	**0.740741**	**xN2-dry**	**0.839979**
	xCO2	0	0	0	**0.088608**	**xCO2-dry**	**0.100478**
	xH2O	0	0	0	**0.118143**		
Frac. xs	*0.2*						
f C3H8	*0.9*						
extent	90						